NAG Library Routine Document

F07WDF (DPFTRF)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07WDF (DPFTRF) computes the Cholesky factorization of a real symmetric positive definite matrix stored in Rectangular Full Packed (RFP) format.

2 Specification

```
SUBROUTINE F07WDF (TRANSR, UPLO, N, AR, INFO)
INTEGER             N, INFO
REAL (KIND=nag_wp) AR(N*(N+1)/2)
CHARACTER(1)        TRANSR, UPLO
```

The routine may be called by its LAPACK name ***dpftrf***.

3 Description

F07WDF (DPFTRF) forms the Cholesky factorization of a real symmetric positive definite matrix A either as $A = U^{\mathrm{T}}U$ if UPLO = 'U' or $A = LL^{\mathrm{T}}$ if UPLO = 'L', where U is an upper triangular matrix and L is a lower triangular, stored in RFP format. The RFP storage format is described in Section 3.3.3 in the F07 Chapter Introduction.

4 References

Demmel J W (1989) On floating-point errors in Cholesky *LAPACK Working Note No. 14* University of Tennessee, Knoxville http://www.netlib.org/lapack/lawnspdf/lawn14.pdf

Gustavson F G, Waśniewski J, Dongarra J J and Langou J (2010) Rectangular full packed format for Cholesky's algorithm: factorization, solution, and inversion *ACM Trans. Math. Software* **37, 2**

5 Parameters

1: TRANSR – CHARACTER(1) *Input*

On entry: specifies whether the RFP representation of A is normal or transposed.

TRANSR = 'N'
The matrix A is stored in normal RFP format.

TRANSR = 'T'
The matrix A is stored in transposed RFP format.

Constraint: TRANSR = 'N' or 'T'.

2: UPLO – CHARACTER(1) *Input*

On entry: specifies whether the upper or lower triangular part of A is stored.

UPLO = 'U'
The upper triangular part of A is stored, and A is factorized as $U^{\mathrm{T}}U$, where U is upper triangular.

UPLO = 'L'
The lower triangular part of A is stored, and A is factorized as LL^{T}, where L is lower triangular.

Constraint: UPLO = 'U' or 'L'.

3: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

4: AR(N $\times$ (N + 1)/2) – REAL (KIND=nag_wp) array *Input/Output*

On entry: the upper or lower triangular part (as specified by UPLO) of the n by n symmetric matrix A, in either normal or transposed RFP format (as specified by TRANSR). The storage format is described in detail in Section 3.3.3 in the F07 Chapter Introduction.

On exit: if INFO $= 0$, the factor U or L from the Cholesky factorization $A = U^{\mathrm{T}}U$ or $A = LL^{\mathrm{T}}$, in the same storage format as A.

5: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO $= -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

The leading minor of order $\langle value \rangle$ is not positive definite and the factorization could not be completed. Hence A itself is not positive definite. This may indicate an error in forming the matrix A. There is no routine specifically designed to factorize a symmetric matrix stored in RFP format which is not positive definite; the matrix must be treated as a full symmetric matrix, by calling F07MDF (DSYTRF).

7 Accuracy

If UPLO = 'U', the computed factor U is the exact factor of a perturbed matrix $A + E$, where

$$|E| \leq c(n)\epsilon \left|U^{\mathrm{T}}\right||U|,$$

$c(n)$ is a modest linear function of n, and ϵ is the ***machine precision***.

If UPLO = 'L', a similar statement holds for the computed factor L. It follows that $\left|e_{ij}\right| \leq c(n)\epsilon\sqrt{a_{ii}a_{jj}}$.

8 Parallelism and Performance

F07WDF (DPFTRF) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F07WDF (DPFTRF) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations is approximately $\frac{1}{3}n^3$.

A call to F07WDF (DPFTRF) may be followed by calls to the routines:

F07WEF (DPFTRS) to solve $AX = B$;

F07WJF (DPFTRI) to compute the inverse of A.

The complex analogue of this routine is F07WRF (ZPFTRF).

10 Example

This example computes the Cholesky factorization of the matrix A, where

$$A = \begin{pmatrix} 4.16 & -3.12 & 0.56 & -0.10 \\ -3.12 & 5.03 & -0.83 & 1.18 \\ 0.56 & -0.83 & 0.76 & 0.34 \\ -0.10 & 1.18 & 0.34 & 1.18 \end{pmatrix},$$

and is stored using RFP format.

10.1 Program Text

```
    Program f07wdfe

!     F07WDF Example Program Text

!     Mark 25 Release. NAG Copyright 2014.

!     .. Use Statements ..
      Use nag_library, Only: dpftrf, dtfttr, nag_wp, x04caf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: i, ifail, info, k, lar1, lda, lenar, &
                                          n, q
      Character (1)                    :: transr, uplo
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: a(:,:), ar(:)
!     .. Executable Statements ..
      Write (nout,*) 'F07WDF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, uplo, transr

      lenar = n*(n+1)/2
      lda = n
      Allocate (ar(lenar),a(lda,n))

!     Setup notional dimensions of RFP matrix AR
      k = n/2
      q = n - k
      If (transr=='N' .Or. transr=='n') Then
        lar1 = 2*k + 1
      Else
        lar1 = q
      End If

!     Read an RFP matrix into array AR
      Do i = 1, lar1
        Read (nin,*) ar(i:lenar:lar1)
      End Do

!     Factorize A
!     The NAG name equivalent of dpftrf is f07wdf
```

```
      Call dpftrf(transr,uplo,n,ar,info)

      Write (nout,*)
      Flush (nout)
      If (info==0) Then

!       Convert factor to full array form, and print it
!       The NAG name equivalent of dtfttr is f01vgf
        Call dtfttr(transr,uplo,n,ar,a,lda,info)
        ifail = 0
        Call x04caf(uplo,'Nonunit',n,n,a,lda,'Factor',ifail)

      Else
        Write (nout,*) 'A is not positive definite'
      End If

    End Program f07wdfe
```

10.2 Program Data

```
F07WDF Example Program Data
  4       'L'      'N' : n, uplo, transr
  0.76    0.34
  4.16    1.18
 -3.12    5.03
  0.56   -0.83
 -0.10    1.18         : RFP matrix AR
```

10.3 Program Results

```
 F07WDF Example Program Results

 Factor
             1          2          3          4
 1      2.0396
 2     -1.5297     1.6401
 3      0.2746    -0.2500     0.7887
 4     -0.0490     0.6737     0.6617     0.5347
```